1. L'espace propre associé à 2 est de dimension 1 et non 3 donc $u$ n'est pas diagonalisable.
2. $v1-(3/2)v2+(1/2)v3=0$ donc la famille $(v1,v2,v3)$ est liée et $u(v1)=v2$, $u(v2)=-2v1+3v2$ : le plan $Vect(v1,v2)$ est stable par $u$, on a la matrice de $u_{|F}$ et comme prévu, son polynôme caractéristique divise $\chi_u$.
   $\chi_{u_{|F}}$ est scindé à racines simples donc $u_{|F}$ est diagonalisable.
3. $-4w1+8w2-5w3+w4=0$ donc la famille $(w1,w2,w3,w4)$ est liée et $u(w1)=w2$, $u(w2)=w3$, $u(w3)=4w1-8w2+5w3$ : l'espace $G$ est stable par $u$ et on a la matrice de $u_{|G}$ ; 2 est valeur propre double associée à un espace propre de dimension 1 donc $u_{|G}$ n'est pas diagonalisable.
4. Soit $uu=u-2Id$ ; $\ker uu$ est de dimension 1 et $\ker(uu)^2$ est de dimension 2. Soit $x2\in\ker(uu)^2\setminus\ker uu$ et $x1=(uu)(x2)$.
   $u(x1)=(uu)^2(x2)+2x1=2x1$ et $u(x2)=(uu)(x2)+2x2=x1+2x2$ donc $H=Vect(x1,x2)$ et stable par $u$, la matrice de $u_{|H}$ dans $(x1,x2)$ est $\begin{pmatrix}2&1\\0&2\end{pmatrix}$ et $\chi_{u_{|H}}=(X-2)^2$.

O19-C07

```
> restart:
> with(LinearAlgebra):
> a:=Matrix([[-23,10,3,-11],[314,-126,-39,139],[-426,174,56,-187],
  [225,-92,-29,100]]);
```

$$a := \begin{bmatrix} -23 & 10 & 3 & -11 \\ 314 & -126 & -39 & 139 \\ -426 & 174 & 56 & -187 \\ 225 & -92 & -29 & 100 \end{bmatrix}$$

```
> u:=<1,2,3,4>;
```

$$u := \begin{bmatrix} 1 \\ 2 \\ 3 \\ 4 \end{bmatrix}$$

```
> a.u;
```

$$\begin{bmatrix} -38 \\ 501 \\ -658 \\ 354 \end{bmatrix}$$

```
> P:=CharacteristicPolynomial(a,X);
```

$$P := X^4 - 7\,X^3 + 18\,X^2 - 20\,X + 8$$

```
> Eigenvectors(a);
```

$$\begin{bmatrix} 1 \\ 2 \\ 2 \\ 2 \end{bmatrix}, \begin{bmatrix} \frac{-10}{91} & 1 & 0 & 0 \\ \frac{128}{91} & \frac{-25}{2} & 0 & 0 \\ \frac{-173}{91} & 17 & 0 & 0 \\ 1 & -9 & 0 & 0 \end{bmatrix}$$

```
> v1:=<-8,103,-139,73>:v2:=a.v1:v3:=a.v2:
> b:=Matrix([v1,v2,v3]);
```

$$b := \begin{bmatrix} -8 & -6 & -2 \\ 103 & 78 & 28 \\ -139 & -105 & -37 \\ 73 & 55 & 19 \end{bmatrix}$$

```
> NullSpace(b);
```

$$\left\{ \begin{bmatrix} 1 \\ \frac{-3}{2} \\ \frac{1}{2} \end{bmatrix} \right\}$$

```
> aF:=Matrix([[0,-2],[1,3]]);
```

$$aF := \begin{bmatrix} 0 & -2 \\ 1 & 3 \end{bmatrix}$$

```
> PF:=CharacteristicPolynomial(aF,X);
```

$$PF := X^2 - 3\,X + 2$$

```
> Eigenvectors(aF);
```

$$\begin{bmatrix} 1 \\ 2 \end{bmatrix}, \begin{bmatrix} -2 & -1 \\ 1 & 1 \end{bmatrix}$$

```
> w1:=<2,-13,17,-9>:w2:=a.w1:w3:=a.w2:w4:=a.w3:
> c:=Matrix([w1,w2,w3,w4]);
```

$$c := \begin{bmatrix} 2 & -26 & -102 & -294 \\ -13 & 352 & 1332 & 3792 \\ 17 & -479 & -1811 & -5155 \\ -9 & 253 & 957 & 2725 \end{bmatrix}$$

```
> NullSpace(c);
```

$$\left\{ \begin{bmatrix} -4 \\ 8 \\ -5 \\ 1 \end{bmatrix} \right\}$$

```
> aG:=Matrix([[0,0,4],[1,0,-8],[0,1,5]]);
```

$$aG := \begin{bmatrix} 0 & 0 & 4 \\ 1 & 0 & -8 \\ 0 & 1 & 5 \end{bmatrix}$$

```
> PG:=CharacteristicPolynomial(aG,X);
> Eigenvectors(aG);
```

$$PG := X^3 - 5\,X^2 + 8\,X - 4$$

$$\begin{bmatrix} 2 \\ 2 \\ 1 \end{bmatrix}, \begin{bmatrix} 2 & 0 & 4 \\ -3 & 0 & -4 \\ 1 & 0 & 1 \end{bmatrix}$$

```
> aa:=a-2*IdentityMatrix(4);
```

$$aa := \begin{bmatrix} -25 & 10 & 3 & -11 \\ 314 & -128 & -39 & 139 \\ -426 & 174 & 54 & -187 \\ 225 & -92 & -29 & 98 \end{bmatrix}$$

```
> K:=[op(NullSpace(aa^2))];x:=K[2];
```

$$K := \left[ \begin{bmatrix} 0 \\ \frac{3}{2} \\ -2 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ -1 \\ 0 \end{bmatrix} \right]$$

$$x := \begin{bmatrix} 1 \\ 1 \\ -1 \\ 0 \end{bmatrix}$$

```
> d:=Matrix([x,a.x,a.a.x]);NullSpace(d);
```

$$d := \begin{bmatrix} 1 & -16 & -68 \\ 1 & 227 & 904 \\ -1 & -308 & -1228 \\ 0 & 162 & 648 \end{bmatrix}$$

$$\left\{ \begin{bmatrix} 4 \\ -4 \\ 1 \end{bmatrix} \right\}$$

```
> aH:=Matrix([[0,-4],[1,4]]);
```

$$aH := \begin{bmatrix} 0 & -4 \\ 1 & 4 \end{bmatrix}$$

```
> PH:=CharacteristicPolynomial(aH,X);
> Eigenvectors(aH);
```

$$PH := X^2 - 4\,X + 4$$

$$\begin{bmatrix} 2 \\ 2 \end{bmatrix}, \begin{bmatrix} -2 & 0 \\ 1 & 0 \end{bmatrix}$$

```
>
```